

# Eclipse Realtime Scheduling Analysis Plugin

Gustavo Hidalgo

Advisors:  
Emmanuel Groulleau  
Yassine Ouhammou

Laboratoire d'Informatique et d'Automatique pour les Systèmes  
École Nationale Supérieure de Mécanique et d'Aérotechnique  
France  
July 2015

# Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Eclipse Modeling</b>	<b>2</b>
<b>3</b>	<b>Meta Model</b>	<b>3</b>
3.1	Complex Validation . . . . .	7
<b>4</b>	<b>Implementation</b>	<b>9</b>
4.1	Main Project . . . . .	9
4.2	Edit . . . . .	10
4.3	Editor . . . . .	10
4.4	Sirius Editor . . . . .	10
4.5	Eclipse Extensions . . . . .	11
4.6	Unexpected Behavior . . . . .	12
<b>5</b>	<b>Example</b>	<b>13</b>
5.1	Project Creation . . . . .	13
5.2	Manipulation . . . . .	13
5.3	Analysis . . . . .	15
5.4	Network . . . . .	18
<b>6</b>	<b>Installation</b>	<b>21</b>
6.1	From Source . . . . .	21
6.2	From JARs . . . . .	21
<b>7</b>	<b>Future work</b>	<b>22</b>
<b>8</b>	<b>Acknowledgements</b>	<b>22</b>

# 1 Introduction

This report describes the design and implementation of an Eclipse plug-in for the purpose of designing real-time systems based on the ARINC 653 standard for integrated modular avionics and ARINC 664 for network connections. The plug-in defines a meta-model for a system which can be used by several algorithms to analytically ensure schedulability stability and an upper bound on the network delay for frames sent over a network. The plug-in also has a comprehensive validation suite which prevents users from creating structurally invalid configurations along with a graphical editor for the network definition which greatly helps to describe and debug network configurations.

This document contains hyperlinks to online resources. View it on a computer to follow the links for more information.

## 2 Eclipse Modeling

The Eclipse Modeling Framework is a tool for designing and manipulating models in Eclipse with many facilities for code generation and persistent data manipulation. It is important to be familiar with this framework in order to understand this project. The most important features used are:

1. **Ecore tools** : A graphical meta-model editor
2. **OCLinEcore** : A modified version of the Object Constraint Language used to define structural constraints for an instance of a meta-model
3. **Code Generation** : A large part of the code in the plug-in was generated by Ecore.
4. **Sirius** : A framework for creating graphical DSLs based on Ecore models

These tools are Eclipse plug-ins which must be installed first before modifying the plugin. If you installed the version of Eclipse used for modeling, you will find a button in the toolbar like in figure 1.




Figure 1: Install modeling plugins

This opens a window with many modeling plugins to install, you should install Sirius and EcoreTools.

### 3 Meta Model

The classes in the Ecore meta-model are described in this section along with the OCLinEcore validation steps that are applied.

For reference, the full meta-model is displayed first and the individual components after. Some of




Figure 2: Full UML meta-model

the key entities of the system will be explained, the others have obvious functions.

**System** The root object for an instance of the meta model. This class serves to hierarchically organize the rest of the classes.

Listing 1: System constraints

```
class System {
    property hardware : HardwareResource { composes };
    attribute name : String;
    property software : SoftwareResource { composes };
}
```

**Hardware and Software Resources** These two classes hierarchically organize other model elements.

Listing 2: Hardware and Software constraints

```
class SoftwareResource {
    property executedTasks : Task[*] { ordered composes };
    property containedPartitions : Partition[+] { ordered composes };
    property virtualLinks : VirtualLink[*] { ordered composes };
}
```

```

class HardwareResource {
  property modules : Module[*] { ordered composes };
  property network : Network[?] { composes };
}

```

**Task** A task is a software service that has well defined temporal activation parameters.

Listing 3: Task constraints

```

class Task {
  attribute id : String { id };
  attribute worstCaseExecTime : ecore::EInt = '0';
  attribute implicitDeadline : ecore::EInt;
  attribute priority : ecore::EInt;
  attribute responseTime : ecore::EInt;
  attribute scheduleable : Boolean;
  attribute criticalInstance : ecore::EInt;
  attribute periodicity : Periodicity = 'Periodic';
  attribute characteristicPeriod : ecore::EInt;
  property partition#tasks : Partition;
  invariant PositiveWCET: worstCaseExecTime > 0;
  invariant ExecutionAndDeadlineAllowsCompletion:
    worstCaseExecTime <= implicitDeadline;
  invariant ExecutionAndPeriodAllowsCompletion:
    if (periodicity <> Periodicity::Aperiodic)
      then worstCaseExecTime <= characteristicPeriod
    else true
    endif;
  invariant DeadlineLessThanPeriod: implicitDeadline <= characteristicPeriod;
  invariant PositivePeriod: characteristicPeriod > 0;
}

```

## Constraint Details

**ExecutionAndDeadlineAllowsCompletion** The WCET must be less than or equal to the implicit deadline otherwise the task will never complete in time.

**ExecutionAndPeriodAllowsCompletion** The WCET must be less than or equal to the period of the task. This is only a constraint on tasks that are not *aperiodic* tasks: tasks without a minimum time between arrivals.

**DeadlineLessThanPeriod** The deadline of the task must be less than or equal to the period otherwise multiple instances of the task will exist simultaneously.

**Partition** A partition is a list of time intervals and a period. Tasks are assigned to partitions.

Listing 4: Partition constraints

```

class Partition {
  attribute id : String = '0' { id };
  attribute period : ecore::EDouble[?] = '0' { derived readonly } {
  derivation:
    if (module->oclIsInvalid() or module->oclIsUndefined() or module = null)
      then 0.0
    else self.module.period
    endif;
  }
  property executionIntervals : Interval[*] { ordered composes };
  attribute virtualProcessorUtilization : ecore::EDouble = '0';
  attribute availabilityFactor : ecore::EDouble = '0' { derived } { derivation:
    --check for divide by zero!
    if (period <> 0)
      then executionIntervals->
        collect(i : Interval | i.end - i.start)->sum() / period
    else 0.0
    endif;
  }
}

```

```

}
attribute schedulingAlgorithm : SchedulingAlgorithm;
property tasks#partition : Task[*] { ordered };
property module#partition : Module;
invariant PositivePeriod: period > 0;
invariant AvailabilityFactorLessThanOrEqualToOne: availabilityFactor <= 1;
invariant PeriodSpansIntervals:
let sortedIntervals : Sequence(Interval) = executionIntervals->
sortedBy(start) in
if (sortedIntervals->size() > 1)
then sortedIntervals->last().end <= period
else true
endif;
invariant NonOverlappingIntervals:
if (executionIntervals->size() <= 1)
then true -- Nothing can overlap if there is only one or none!
else
let sortedIntervals : Sequence(Interval) = executionIntervals->
sortedBy(i : Interval | i.start) in
sortedIntervals->
subSequence(1, sortedIntervals->size() - 1)->
forAll(i : Interval |
i.end <= sortedIntervals->at(1+sortedIntervals->indexOf(i)).start)
endif;
}

```

### Constraint Details

**period** The period of a Task is defined to be the period of the module it executes on. This simplifies validating that no partitions overlap within a module.

**availabilityFactor** The fraction of time this partition occupies within a module. This is not the same as the virtual processor utilization which is the fraction of time that any task is executing in a partition.

**PeriodSpansIntervals** The intervals defined in a partition must be wholly contained within  $t = (0, T)$ .

**NonOverlappingIntervals** No two intervals within a partition may overlap. This is achieved by imposing an ordering on the intervals and checking to make sure that no two sequential intervals overlap.

**Interval** An interval is a period of execution for tasks in a partition.

Listing 5: Interval constraints

```

class Interval {
attribute start : ecore::EInt;
attribute end : ecore::EInt;
invariant EndAfterStart: end >= start;
invariant NonZeroLength: end <> start;
}

```

**Network** A network is a physical description of the network hardware, namely switches, connections, and ports on a switch.

Listing 6: Network constraints

```

class Network {
attribute name : String;
property switches : Switch[*] { ordered composes };
attribute networkBandwidth : ecore::EInt = '0';
property connections : Connection[*] { ordered composes };
attribute latency : ecore::EInt = '0';
invariant PositiveNetworkLatency: latency > 0;
invariant PositiveNetworkBandwidth: networkBandwidth > 0;
}

```

**Module** A module is an ‘End system’ in the view of the network and a processing element where software resources execute.

Listing 7: Module constraints

```
class Module
{
  attribute utilization : ecore::EDouble = '0' { derived } { derivation:
    if (partition->size() > 0)
    then partition->collect(p | p.availabilityFactor)->sum()
    else 0.0
    endif;
  }
  attribute id : String;
  attribute period : ecore::EInt;
  property partition#module : Partition[*] { ordered };
  property processors : Processor[+] { ordered composes };
  property modulePorts : EndSystemPort[*] { ordered composes };
  attribute delay : ecore::EInt;
  invariant PositiveDelay: delay > 0;
  invariant NonZeroPeriod: period > 0;
  invariant NonOverlappingPartitions:
    let allIntervals : Sequence(Interval) =
    partition.executionIntervals->flatten()->sortedBy(i : Interval | i.start)
    in if (allIntervals->size() <= 1)
    then true
    else allIntervals->subSequence(1, allIntervals->size() - 1)
    ->forall(i : Interval |
      i.end <= allIntervals->at(1 + allIntervals->indexOf(i)).start)
    endif;
}
}
```

### Constraint details

**utilization** This derived quantity is based on the partitions scheduled on the module. It is the sum of the availability factors of each partition.

**NonOverlappingPartitions** This applies the same overlapping logic as an individual partition’s overlapping check but to all partition assigned to this module.

**Switch** A switch is a network device that helps to connect devices.

Listing 8: Switch constraints

```
class Switch {
  attribute id : String;
  property switchPorts : SwitchPort[+] { ordered composes };
  invariant AtleastTwoSwitchPorts: switchPorts->size() > 1;
}
```

**Virtual Link** A Virtual Link represents a logical connection between one ‘source’ end system and one or more ‘destination’ end systems.

Listing 9: Virtual Link constraints

```
class VirtualLink {
  attribute id : String;
  attribute BAG : ecore::EInt;
  property routes : Route[+] { ordered composes };
  property source : Module;
  property destinations : Module[+] { ordered };
  attribute maxFrameSize : ecore::EInt = '0';
  invariant DestinationsCannotIncludeSource:
    destinations->forall(dest : Module | dest <> source);
  invariant PositiveBAG: BAG > 0;
  invariant PositiveMaxFrameSize: maxFrameSize > 0;
}
```

```

    invariant PathExists;
    invariant RoutesConnectSourceToDestinations;
    invariant NoCycles;
}

```

**Constrain details** See 3.1.

### 3.1 Complex Validation

Some validation could not be expressed in OCLinEcore.

**Network Invariants** There are invariants for a `VirtualLink` that are better expressed through graph algorithms.

Listing 10: VirtualLink graph invariants

```

invariant PathExists;
invariant RoutesConnectSourceToDestinations;
invariant NoCycles;

```

These invariants are expressed in the `fr.ensma.realsimescheduling.util.RealsimeschedulingValidator.java` class as part of the validation code. We used the free and open source Java graph library **JGraphT**. The first constraint `PathExists` verifies that there is atleast one path possible between the source of a virtual link and each of its destinations. We consider all of the ports in the network as nodes and all of the ports of a switch are fully connected. Regular connections are edges. First we build the graph.

```

1  private void buildGraph(fr.ensma.realsimescheduling.System system) {
2      networkGraph = new SimpleGraph<>(DefaultEdge.class);
3      for (Switch switch_ : system.getHardware().getNetwork().getSwitches()) {
4          for (Port port : switch_.getSwitchPorts())
5              networkGraph.addVertex(port);
6          for (Port port : switch_.getSwitchPorts())
7              for (Port port2 : switch_.getSwitchPorts())
8                  // within a switch, all ports are "connected" to each other.
9                  if (port2 != port)
10                     networkGraph.addEdge(port, port2);
11      }
12      for (Module module : system.getHardware().getModules())
13          for (Port port : module.getModulePorts())
14              networkGraph.addVertex(port);
15      for (Connection connection : system.getHardware().getNetwork().getConnections())
16          if (connection.getA() != null && connection.getB() != null)
17              networkGraph.addEdge(connection.getA(), connection.getB());
18  }

```

Then we ask JGraphT to check the connectivity of every virtual link.

```

1  ConnectivityInspector<Port, DefaultEdge> inspector = new ConnectivityInspector<>(networkGraph);
2  boolean pathExists = false;
3  try {
4  all: for (Port sourcePort : virtualLink.getSource().getModulePorts())
5      for (Module destination : virtualLink.getDestinations())
6          for (Port destinationPort : destination.getModulePorts()) {
7              pathExists = inspector.pathExists(sourcePort, destinationPort);
8              if (!pathExists)
9                  break all;
10          }

```



```

11     } catch (Exception e) {
12     e.printStackTrace();
13     }

```

---

The next constraint verifies that the specified routes actually connect the correct modules. This is done by ‘walking’ a route and verifying that the last connection connects one of the destinations. If, at the end, all destinations have been reached, then all of the routes connect all of the destinations and the virtual link is valid.

---

```

1  boolean success = false;
2  List<Module> destinationsHit = new ArrayList<>();
3  destinationsHit.addAll(virtualLink.getDestinations());
4  try {
5      for (Route r : virtualLink.getRoutes()) {
6          Port first = virtualLink.getSource().getModulePorts().stream()
7              .filter(esp -> r.getConnections().contains(esp.getConnection()))
8              .findFirst()
9              .get();
10         Port current = first;
11         List<Connection> allConnections = new ArrayList<>();
12         allConnections.addAll(r.getConnections());
13         while (allConnections.size() != 1) {
14             allConnections.remove(current.getConnection());
15             current = Flow.getOpposite(current);
16             current = ((Switch) (current.eContainer())).getSwitchPorts().stream()
17                 .filter(sp -> allConnections.contains(sp.getConnection()))
18                 .findFirst()
19                 .orElseThrow(() -> new Exception());
20         }
21         success = virtualLink.getDestinations().stream()
22             .anyMatch(module -> {
23                 boolean matches = module.getModulePorts().contains(allConnections.get(0).getB());
24                 destinationsHit.remove(module);
25                 return matches;
26             })
27         || /* OR */
28         virtualLink.getDestinations().stream()
29             .anyMatch(module -> {
30                 boolean matches = module.getModulePorts().contains(allConnections.get(0).getA());
31                 destinationsHit.remove(module);
32                 return matches;
33             });
34     }
35 } catch (Exception e) {
36     e.printStackTrace();
37     success = false;
38 }

```

---

The cycle checker is not implemented. See section 7.

## 4 Implementation

This section describes Java code and Eclipse specific implementation details of the plug in. The plug in is composed of 5 Eclipse projects, 4 generated from the Ecore model and 1 from the Sirius specification. See figure 3 for the list of projects.




Figure 3: Plug in projects

Most of the non-generated code resides in `fr.ensma.realtimescheduling`, specifically the analysis algorithms, visualization classes, and the meta-model interfaces and implementations. This project will be referred to as the **main** project.

The projects `edit`, `editor`, and `test` contain *generated* code for editing the model (operations such as instance creation, or attribute modification), the editor GUI (see figure 5), and test code respectively. The test project is currently not used.

Several Eclipse extensions are also declared in the `plugin.xml` within the main project (see figure 7). These mainly consist of UI extensions to allow for easier manipulation and analysis of a model.

### 4.1 Main Project

The main project contains the analysis algorithms and non-trivial UI logic code for the project. The package structure of this project is given in figure 4.




Figure 4: Main project packages

Several of these packages deserve special attention.

**analysis** This package contains the algorithms and “glue” code between the model and the algorithms.

The `Analyzer.java` class holds the implementations of all the algorithms described in the CORAC-PANDA project: the response time analysis from chapter 2 and the network delay analysis from chapter 3. The code was written attempting to recreate as much as possible the structure and names of the pseudo-code from the report.

The `ModelInterface.java` class contains methods to interface between the UI and the analysis algorithms.

The rest of the classes in the package provide side-effect free functions for manipulating the model or classes to store auxiliary data necessary by the algorithms. For example, the Ecore list implementation `EList` is not modifiable thus it cannot be passed to Java Collections methods such as `Collections.sort()`. Having a list of time-sorted `Intervals` is necessary for the algorithms thus the `PartitionUtils.java` file provides such a method by creating a copy of the original list.

**handlers** This package contains handler definitions for the commands generated by Eclipse. UI actions in Eclipse are handled by a command-handler relationship. More information on this mechanic can be found online.

**views** This package contains the classes used to layout and display data to charts from the model. This project uses the **JFreeChart** library to generate charts like line charts and bar charts. The classes `AbstractLineChart.java` and `AbstractBarChart.java` set up the layout of the charts which is a horizontal layout with a chart on the left and a control component on the right.

**util** This package is generated by Ecore, however it houses the `RealtimeschedulingValidator.java` class which contains the complex validation code that could not be expressed in OCLinEcore. The specifics of the validation are in section 3.

## 4.2 Edit

This project contains generated code for modifying elements of a model instance. The only modifications to this code are found in the `getText()` method of some of the instances which makes the display in the editor more user friendly. See figure 5.

## 4.3 Editor

This project defines the generated classes for a simple model editor which only displays a tree representation of the model.




Figure 5: Model editor in use

## 4.4 Sirius Editor

A view definition project was also created with the intent of making the network definition easier. Editors that this project provides are only available in the runtime instance of Eclipse.

The sirius project defines views to 'view' the network definition as well as the ability to edit it. This is defined in an editor very similar to the editor of the model. Refer to section 4.6 for details about this editor.




Figure 6: Sirius definition

## 4.5 Eclipse Extensions

The plug in provides several custom Eclipse extensions apart from the generated ones.




Figure 7: Plug in extensions

**Menu** Menu extensions add items to existing menus. In this case, I added a menu item to the main Eclipse menu bar (see figure 8). This is indicated by the root **Menu Contribution** element of the extension which indicates that this menu should be added to the menu specified by `menu:org.eclipse.ui.main.menu`.




Figure 8: Menu extension definition and result

**View** Several new views are defined in the project. These are mainly for graphing the information derived from the analysis of the model.




Figure 9: View extension definition and result

**Perspective** One new perspective is declared with the purpose of arranging the model editor and the graphing views into a coherent layout. New perspectives can be defined by first creating an extension of `org.eclipse.ui.perspective` and then of `org.eclipse.ui.perspectiveExtensions`. The first is a manifest to Eclipse that this plug-in defines a new extension; the second allows you to define the layout of the extension from within the `plugin.xml` editor.



**SWT Layout** Eclipse uses the SWT windowing toolkit which allows you to define layouts like any other major windowing toolkit. However, it is important to know that at runtime `Layout` object may be cast into more specific layouts which can causes exceptions. Each `Composite` (a member with children) of a GUI has a `layout` and `layoutData`. The `layoutData` is used by the layout of the *parent* object to arrange the `Composite` while the `layout` of the composite is used to arrange its children. This became an issue in the graphing views because they mix different layout managers and these casts would crash the view at runtime.

12

## 5 Example

This section shows an example model created using the plug-in. I describe how to create it and how to analyze it.

### 5.1 Project Creation

You will first have to create a new empty project in Eclipse.




Figure 11: Create a new project

Then select one of the empty project templates and create a new empty sirius project. Give it a name and create the empty project. Then, right click on the new empty project and go to **New**, then **Other**.




Figure 12: Creation

Here you can search for the **Realtimescheduling** model to create an instance of the model. Select the model and create it with a name. When prompted, make sure to select the root element **System**.




Figure 13: Select model

### 5.2 Manipulation

You should change into the **Realtime Analysis** perspective to maximize your use of the plugin. You should see the editor view appear with a single element that is the **System** object. You will need to define

the model instance by compositional relations first. This can be done by right-clicking on an element and adding children or sibling objects.




Figure 14: Adding children

When you select a model element, the properties editor should change to display the attributes of the element. Here you can change any attribute declared as **changeable** by double clicking on the value field and entering a new value.




Figure 15: Properties

Some of the fields will set referential relations and will ask you to choose from a list of instance objects: for example, if you were to set the Tasks of a partition then a window will open to display all the tasks available. For this example, we will work on a fictitious quadcopter system where the attitude controller tasks have hard realtime constraints. Lets add some tasks and partition definitions.




Figure 16: Defined system

Now we can start probing the system but first we would like to make sure that the system is “Valid” first: that it is structurally sound. We can do that by right-clicking on the **System** object and then clicking on **Validate**.




Figure 17: Validate

If the model successfully validates, we should see an “Success” message. But if it does not, we will see errors on the objects that did not pass validation.




Figure 18: Validation messages

We can quickly fix the errors and see the expected success prompt. Once the model has successfully validated, we can safely perform analysis on it.

### 5.3 Analysis

Open the **Analysis** menu in the main toolbar to see the possible options.




Figure 19: Analysis menu

We have not defined a network yet so we are constrained to analyzing the tasks schedulability behavior. Click on **Analysis** then **Response Time**. This will run the analysis algorithm described in this paper and display the results per-partition and per-task. Here we can see that the **Motors** tasks will miss its deadline. We can also take a look at the **Response Time** view to see a bar chart of the worst case response time for all tasks in the system.






Figure 20: Analysis results

We have a problem with the Motors missing their deadline. We have several options as designers: we can decrease the execution time of the tasks, we can increase its period, we can redistribute the time in the partitions, or we can move it into a different partition. The first and the last options are the most difficult: it might not be possible to write faster code and it may be unsafe to move such a critical task to a partition with lower priority tasks. The other two are tractable: we will add more time to the “Fast” partition and increase the period of the Motors task. Rerunning the analysis will show us that all tasks are now schedulable.




Figure 21: Scheduling success

Now that a stable configuration is achieved, we can start thinking about adding new functions to the system. For example, we would like to add an altitude control task to run alongside the PID controllers. We define a new task with  $T, D = 200$  and  $C = 50$ . Unfortunately this tasks becomes unschedulable.




Figure 22: Scheduling failure

We can try to make it schedulable through trial and error but the plugin can provide some information

to help us make these decisions. In the **Partition Functions** view, we can see that the **Least Supply** function curve for the “Fast” partition has a value of 60 for  $t = 200$ . This means that for all intervals of time of length 60, the minimum amount of supply available is 60. This task requires 50 supply to complete but it must share that with the other PID controller’s tasks. Therefore by examining the least-supply function and the other tasks in the system, we can estimate that this task is unschedulable because for its period (200), there is barely enough time to complete it and in fact the other tasks are cannibalizing the time for the altitude controller. If we want a safe cushion of time for this task, we can look at larger values for the period in the least-supply curve. We can also attempt to make this task a higher priority task manually by assigning it a higher priority as a property. However, this has disastrous results.




Figure 23: Priority changes do not fix the problem

Notice that tasks in the “Slow” partition are unaffected by our changes to the “Fast” partition. This isolation is good for the designer but also for the system where tasks will not harm the execution of other partition if something goes wrong. We will force the system to be schedulable by reducing the execution time of the Altitude controller but remember that there are other means to do so. We can look at the virtual processor utilization of a partition to help us decide where to place new tasks. This function is in the **Analysis** menu.




Figure 24: Virtual processor utilization

The “Fast” partition has a relatively high VPU: around 75%. The slow partition’s low VPU makes it a candidate for adding more tasks because the partition is only being utilized for 20% of the time it is executing. There is also a VPU view chart available. Let’s say that we want the drone to provide mobile wifi access. We define some new tasks.




Figure 25: New wifi tasks

Immediately we see that these new tasks are unschedulable, but only in the “Slow” partition!




Figure 26: Local partition effects

Let us look to the least supply function for guidance. The LSF for the “Slow” partition has a value of 0 at  $t = 50$  which is the period of the wifi tasks.




Figure 27: Least supply for “Slow”

This means that no task with a period less than  $t = 130$  will ever complete in the slow partition because there is a critical instance where the task will not get to execute for 130 time units. This lines up with the times  $(0, 130)$  which the “Fast” partition claimed. Knowing this, we must make these tasks have longer periods. Once we make them schedulable we can then see that this increases the VPU of the “Slow” partition.




Figure 28: Increased VPU

## 5.4 Network

Now we will define a network component, although quadcopters do not communicate with motors based on packet switched networks. First, open the graphical editor by first right clicking on the project and selecting **Viewpoint Selection**. You should see a single option named ‘Design’, select that. Right click on the project again but now select **Create Representation** and choose the ‘Network’ representation with the System object as the root.




Figure 29: Creating the graphical editor

Create a few more modules to represent the motors of the quadcopter. You should then see a few modules in the editor. If you do not see anything, you may need to enable the layers which you can do with the toolbar button of the editor.




Figure 30: Viewing the network

Before you continue editing the Network, you should make sure that there is an instance of a Network object in the model. ***This is very important!***




Figure 31: IMPORTANT

Now you can define your network using the tools on the right side of the graphical editor. The model will update with your changes when you save the file. Add a switch with 5 ports and add module ports to all the modules. Also connect them up. Connections can only be made between two distinct ports.




Figure 32: Defined network

You may also use the virtual link definition tool by enabling the Virtual Link layer. This allows you to define virtual links from within the editor but you will still need to define the connections from the tree editor.




Figure 33: Defining virtual links

If you make a mistake in designing your network, the validator should catch it. Now you may run the network delay analysis options from the **Analysis** menu.

## 6 Installation

### 6.1 From Source

Move the 5 projects into your Eclipse workspace. You can then highlight the 5 projects, right-click on one of them, select **Run As** then **Eclipse Application**.

### 6.2 From JARs

You can drag and drop the JARs into your Eclipse plugin directory. You should then see the project available as a new project along with the views and the perspective.

## 7 Future work

This section describes what future work needs to be done to improve the project.

1. Add units to all attributes that need them. for example, the worst case execution time of tasks could be in milliseconds or microseconds. Consistency is important in this because the code would have to be modified with unit conversions.
2. Better Sirius diagrams with more functionality. Although it is possible to edit the model entirely from the tree editor, it is error prone and difficult. This would require serious knowledge of sirius best practices.
3. More analysis algorithms.
4. Cycle detection in the network is a little harder because JGraphT does not provide an undirected graph cycle checker. I would like to avoid writing the algorithm but apparently **it is not so hard**.
5. The analysis algorithms expect that the model is valid. This is too optimistic.
6. Currently the `ModelInterface` class receives the model once it is validated. There should be a better way to reference the model besides waiting for the user to validate the model.

## 8 Acknowledgements

I want to thank Emmanuel Groulleau for giving me the opportunity to come to France and perform this internship. These last two months have been very rewarding and I hope to visit again soon. I also want to thank Yassine Ouhammou for giving me direction when it comes to designing this plug-in and for helping me navigate the world of real-time scheduling analysis.